Визуализация решений обыкновенных дифференциальных уравнений

Eugene Strakhov

Please report bugs to strakhov.e.m@onu.edu.ua

1 Модель динамики изменения численности популяции

Рассмотрим некоторую популяцию, имеющую достаточно пищи и ограниченной от влияния хищников. Обозначим $N(t)$ число особей в момент времени t. Тогда естественно предположить, что скорость роста числа особей пропорциональна их количеству с некоторым положительным коэффициентом r (т. е. в каждый момент времени r-я часть особей даёт потомство):

$$\frac{\mathrm{d}N(t)}{\mathrm{d}t} = r \cdot N(t).$$

Это дифференциальное уравнение с разделяющимися переменными имеет решение $N(t) = e^{rt}$, т. е. мы получаем, что численность популяции растёт экспоненциально. На практике это не так, ведь особи конкурируют между собой за ресурсы на ограниченной территории. Для того, чтобы модель соответствовала реальной ситуации, введём поправочный множитель $\left(1 - \frac{N}{K}\right)$, где $K > 0$:

$$\frac{\mathrm{d}N}{\mathrm{d}t} = rN\left(1 - \frac{N}{K}\right). \tag{1}$$

При значениях N, близких к нулю, этот коэффициент близок к единице, следовательно, имеем рост численности, близкий к экспоненциальному. Если N близко к K, то поправка близка к нулю и рост популяции останавливается (в правой части уравнения стоит нуль, т. е. $N = \mathrm{const}$). Если же N превышает K, правая часть уравнения становится отрицательной и численность популяции начинает убывать. Поэтому параметр K можно интерпретировать как предельную ёмкость среды — число особей, при котором ещё возможен прирост численности популяции.

Уравнение (1) называется **логистическим уравнением**, или **уравнением Ферхюльста**[1]. Интересно, что данное уравнение Ферхюльст получил, рассматривая модель роста численности населения.

```
t = var('t')
N = function('N')(t)
r = 0.2
K = 10
deq = diff(N, t) == r*N*(1-N/K)
N0 = 15
f = desolve(deq, N, ics=[0, N0], show_method=True, contrib_ode=True)
show(f)
```

$$[-5\,\log\left(N\left(t\right) - 10\right) + 5\,\log\left(N\left(t\right)\right) = t + 5\,\log\left(15\right) - 5\,\log\left(5\right), \texttt{separable}]$$

Решение найдено в неявном виде, поэтому его график возможно построить только как `implicit_plot`. Также построим поле направлений движения — `plot_slope_field`.

```
y = var('y')
ff = f[0].log_simplify().substitute(N==y)
show(ff)
ip = implicit_plot(ff, (t,0,20), (y,0,20), color='red', linewidth=3)
psf = plot_slope_field(r*y*(1-y/K), (x,0,20), (y,0,20), headaxislength=3, headlength=3)
show(ip+psf, aspect_ratio=0.5, axes_labels=['time', 'population'])
```

```
...
```

Из рисунка видно, что численность популяции стремится к 10 (ёмкость среды).

Далеко не всегда удаётся решить дифференциальное уравнение аналитически. Воспользуемся численным решением.

[1] По фамилии впервые исследовавшего его бельгийского математика Pierre François Verhulst (1804–1849).

```
pn = desolve_rk4(deq, N, ics=[0, N0], output='slope_field', end_points=[0, 20]) # опции out
show(pn, axes_labels=['time', 'population'])
```

```
...
```