Handouts/Lecture 6 (Mar 29)/Lecture 6: Differential Equations/Diff Equations.sagews

Author **Eugene Strakhov**

Date **2017-03-28T22:43:37**

Project **bf655f90-eca0-470e-9c17-2df7d93ad139**

Location Handouts/Lecture 6 (Mar 29)/Lecture 6: Differential Equations/Diff Equations.sagews

Original file Diff Equations.sagews

Аналитическое решение дифференциальных уравнений

```
desolve(de, dvar) # аналитическое решение д. у. 1-го или 2-го порядка, используя Maxima
desolve(de, dvar, ics)
desolve(de, dvar, ics, ivar)
desolve_system(de_sys, dvars) # аналитическое решение системы д. у. 1-го порядка
desolve_system(de_sys, dvars, ics)
desolve_system(de_sys, dvars, ics, ivar)
```

Пример 1. Найдём общее решение уравнения $y'' = x^2$.

```
y = function('y')(x) # объявляем неизвестную функцию
de = diff(y, x, 2) == x^2 # задаём само уравнение
show(de)
show(desolve(de, y)) # ответ с константами
```

$$\frac{\partial^2}{(\partial x)^2} y(x) = x^2$$

$$\frac{1}{12} x^4 + K_2 x + K_1$$

Пример 2. Найдём общее решение уравнения $y'' = ax + b$, где a и b — некоторые параметры.

В данном случае нам придётся явно указать независимую переменную — **ivar** (**i**ndependent **var**iable).

```
y = function('y')(x)
a, b = var('a b')
de = diff(y, x, 2) == a*x + b
sol = desolve(de, y, ivar=x)
show(sol)
sol.variables() # обратите внимание, как можно обратиться к константе
```

$$\frac{1}{6} a x^3 + \frac{1}{2} b x^2 + K_2 x + K_1$$

```
(_K1, _K2, a, b, x)
```

Пример 3. Решим задачу Коши $y' - \frac{y}{x} = x \sin x$, $y\left(\frac{\pi}{2}\right) = 1$.

При решении задачи Коши нам понадобится параметр **ics** — начальные условия (**i**nitial **c**ondition**s**). Он задаётся в виде списка `[a, y(a), y'(a), ...]` (длина зависит от порядка уравнения).

```
y = function('y')(x)
de = diff(y,x) - y/x == x*sin(x)
sol = desolve(de, y, [pi/2, 1], show_method=True) # можно увидеть тип уравнения (т. е. метод решени
show(sol) # решением теперь будет список: sol[0] - само решение, s[1] - метод
show(sol[0](x=pi/2)) # проверим, всё ли правильно
```

$$\left[-\frac{\pi x \cos(x) - 2x}{\pi}, \text{linear}\right]$$

1

Пример 4. Решим систему

$$\begin{cases} \dot{x}_1 = x_1 + x_2 + t, \\ \dot{x}_2 = -4x_1 - 3x_2 + 2t \end{cases}$$

при начальных условиях $x_1(0) = 1$, $x_2(0) = 0$.

```
t = var('t')
x1 = function('x_1')(t)
x2 = function('x_2')(t)
de1 = diff(x1,t) == x1+x2+t
de2 = diff(x2,t) == -4*x1-3*x2+2*t
sol = desolve_system([de1,de2], [x1,x2], ivar=t, ics=[0,1,0]) # обратите внимание на ics
show(sol) # решение будет в виде списка выражений
# Выведем решения отдельно
for s in sol :
    show(s.right())
```

$$\left[x_1\left(t\right) = 6\,te^{\left(-t\right)} + 5\,t + 10\,e^{\left(-t\right)} - 9\ ,\ x_2\left(t\right) = -12\,te^{\left(-t\right)} - 6\,t - 14\,e^{\left(-t\right)} + 14\ \right]$$

$$6\,te^{\left(-t\right)} + 5\,t + 10\,e^{\left(-t\right)} - 9$$

$$-12\,te^{\left(-t\right)} - 6\,t - 14\,e^{\left(-t\right)} + 14$$

Замечание. Если дифференциальное уравнение не решается с первого раза (или ответ получен в неявном виде), можно попробовать указать дополнительный параметр `contrib_ode=True`. Это позволит решить уравнения Лагранжа, Клеро, Риккати и некоторые другие типы уравнений.

Численное решение дифференциальных уравнений

```
desolve_rk4(...) # уравнения первого порядка
desolve_system_rk4(...) # системы первого порядка
desolve_mintides(...) # системы первого порядка (via TIDES)
desolve_odeint(...) # системы первого порядка (via scipy)
```

См. отдельные файлы.

generated 2017-03-28T22:43:37 on SageMathCloud